Problem:
Score: 300 points

Problem Statement

In the Ancient Kingdom of Snuke, there was a pyramid to strengthen the authority of Takahashi, the president of AtCoder Inc.

The pyramid had center coordinates (C_X, C_Y) and height H. The altitude of coordinates (X, Y) is max(H - |X - C_X| - |Y - C_Y|, 0).

Aoki, an explorer, conducted a survey to identify the center coordinates and height of this pyramid. As a result, he obtained the following information:

C_X, C_Y was integers between 0 and 100 (inclusive), and H was an integer not less than 1.

Additionally, he obtained N pieces of information. The i-th of them is: "the altitude of point (x_i, y_i) is h_i."

This was enough to identify the center coordinates and the height of the pyramid. Find these values with the clues above.

Constraints

N is an integer between 1 and 100 (inclusive).

x_i and y_i are integers between 0 and 100 (inclusive).

h_i is an integer between 0 and 10^9 (inclusive).

The N coordinates (x_1, y_1), (x_2, y_2), (x_3, y_3), ..., (x_N, y_N) are all different.

The center coordinates and the height of the pyramid can be uniquely identified.

Input

Input is given from Standard Input in the following format:

N
x_1 y_1 h_1
x_2 y_2 h_2
x_3 y_3 h_3
:
x_N y_N h_N

Output

Print values C_X, C_Y and H representing the center coordinates and the height of the pyramid in one line, with spaces in between.

Sample Input 1

4
2 3 5
2 1 5
1 2 5
3 2 5

Sample Output 1

2 2 6

In this case, the center coordinates and the height can be identified as (2, 2) and 6.

Sample Input 2

2
0 0 100
1 1 98

Sample Output 2

0 0 100

In this case, the center coordinates and the height can be identified as (0, 0) and 100.

Note that C_X and C_Y are known to be integers between 0 and 100.

Sample Input 3

3
99 1 191
100 1 192
99 0 192

Sample Output 3

100 0 193

In this case, the center coordinates and the height can be identified as (100, 0) and 193.

Solution in Haskell (GHC 7.10.3) (judge verdict: Accepted)

import Control.Monad (replicateM)

type Pos = (Int, Int)
type Height = Int

diff :: Pos -> Pos -> Height
diff (x1, y1) (x2, y2) = abs (x1 - x2) + abs (y1 - y2)

veridation :: Pos -> Height -> Pos -> Height -> Bool
veridation x h c ch = h == max 0 (ch - diff c x)

candidates :: Pos -> Height -> [(Pos, Height)]
candidates x h = map (\p -> (p, h + d p)) [(x', y') | x' <- [0..100], y' <- [0..100]]
  where d = diff x

takeSample :: [(Pos, Height)] -> (Pos, Height)
takeSample = head . filter (\(_, h) -> 0 < h)

parse :: IO (Pos, Height)
parse = do
  [x, y, h] <- map read . words <$> getLine
  return ((x, y), h)

fil' :: (Pos, Height) -> [(Pos, Height)] -> [(Pos, Height)]
fil' (x,h) cs = filter (uncurry (veridation x h)) cs

fil :: [(Pos, Height)] -> [(Pos, Height)] -> [(Pos, Height)]
fil ss cs = foldr fil' cs ss

main = do
  n <- read <$> getLine :: IO Int
  samples <- replicateM n parse
  let (sx, sh) = takeSample samples
      cs = candidates sx sh
      ((x,y), h) = head $ fil samples cs
  putStrLn $ show x ++ ' ' : show y ++ ' ' : show h